

CRAPS 2
INSTRUCTIONS

1 TO 4 PLAYERS CAN PLAY. EACH PLAYER STARTS OUT WITH \$500.

AFTER THE PLAYER PLACES HIS BET, THE BALLY WILL SAY:"ROLL;" HIT ANY KEY ON THE KEYPAD TO GET THE DICE TO START ROLLING.

THE RULES ARE THE SAME AS IN LAS VEGAS.

ON THE FIRST ROLL;
YOU LOSE ON 2, 3, OR 12
YOU WIN ON 7 OR 11

IF YOU DON'T ROLL ANY OF THESE NUMBERS ON YOUR FIRST ROLL THEN YOU HAVE TO ROLL AGAIN. YOU HAVE TO ROLL THE SAME NUMBER AS YOU DID ON YOUR FIRST ROLL TO WIN. YOU LOSE IF YOU ROLL A 7.

CRAPS 2

```

1 .CRAPS
2 .VERSION 2
3 CLEAR
4 PRINT
5 BC=195;FC=7
6 F=0;I=0
8 NT=3
10 INPUT "HOW MANY PLAYERS?"P
13 IF P>4 GOTO 10
14 IF P<1 GOTO 10
15 FOR M=1 TO P
20 @(M)=500
30 NEXT M
40 FOR A=1 TO P
41 CLEAR
42 IF @(A)=0 GOTO 280
43 PRINT
45 PRINT "PLAYER",#2,A; INPUT "WHAT IS YOUR BET?"B
47 IF B>@(A) PRINT "YOU ONLY HAVE ",#2,@(A), " Dollars"; GOTO 43
50 PRINT; PRINT "ROLL:"
53 NT=10
54 PRINT "135*105*10000";NT=3
55 Z=KP
57 GOSUB 70
58 GOTO 225
70 FOR Z=1 TO RND (10)+5
73 CLEAR
75 D=RND(6);C=RND(6)
80 T=RND(10)-30;L=RND(10)-5
85 BOX T,L,14,14,1
90 X=T+RND(15)+5; Y=L+RND(15)
95 BOX X,Y,14,14,1
100 NEXT Z
110 BOX T,L,14,14,1; BOX X,Y,14,14,1
119 CX=T;CY=L
120 PRINT #1,C
135 GOTO 200
200 CX=X;CY=Y
205 PRINT #1,D
215 D=D+C
217 CY=40;PRINT
220 PRINT " YOU ROLLED A:",#2,D
221 RETURN
225 IF D=7 PRINT "YOU WIN!";GOTO 230
226 IF D=11 PRINT "YOU WIN!";GOTO 230
227 GOTO 240
230 NT=10;PRINT;PRINT "3050034050000";NT=3
233 @(A)=@(A)+B
234 FOR O=1 to 800;NEXT O;CLEAR;CX=-80;CY=0
235 PRINT "PLAYER #",#2,A,
237 PRINT #6,@(A), " DOLLARS"
238 FOR U=1 to 800; NEXT U;CLEAR;GOTO 280
240 IF D=2 PRINT "YOU LOSE";GOTO 250
241 IF D=3 PRINT "YOU LOSE";GOTO 250
242 IF D=12 PRINT "YOU LOSE";GOTO 250
245 PRINT "YOUR POINT IS",#3,D
246 F=D
247 GOTO 330
250 @(A)=@(A)-B;IF@(A)<1 PRINT "YOU'RE OUT OF MONEY";GOTO 260

```

```
255 GOTO 234
260 @(A)=0
265 Q=0
270 FOR Z=1 to P
272 IF @(Z)< 1 Q=Q+1
275 NEXT Z
277 IF Q=P GOTO 308
280 NEXT A
285 CLEAR
290 PRINT;PRINT;PRINT "SCORE:"
293 PRINT
295 FOR Z=1 TO P
297 IF @(Z)>30000 PRINT "PLAYER #",#2,Z," BROKE THE BANK!";I=1; GOTO 302
300 PRINT "PLAYER #",#2,Z,#6,@(Z)," DOLLARS"
302 PRINT
305 NEXT Z
306 FOR U=1 TO 800; NEXT U;IF I=1 GOTO 380
307 GOTO 40
308 FOR U=1 TO 700; NEXT U;CLEAR;CX=-80; CY=0
310 PRINT "THAT'S THE END OF THE GAME ALL PLAYERS ARE OUT OF MONEY"
315 NT=10
320 PRINT "123456700 135*105*10000"
325 GOTO 400
330 PRINT; PRINT" ROLL:"
335 FOR U=1 to 800; NEXT U
350 GOSUB 70
355 IF D=7 PRINT "YOU LOSE!";GOTO 250
360 IF D=F PRINT "YOU MADE YOUR POINT";GOTO 230
370 GOTO 330
380 CLEAR;CY=0;PRINT "THE BANK IS OUT OF MONEY!"
385 FOR Z=1 to 700; NEXT Z;CLEAR
390 CX=-20;CY=0;PRINT #1,"GAME OVER"
400 NT=3
```